



## **Progetto di Ricerca Operativa**

*Ottimizzazione dei Costi per i Servizi di Trasporto Ordinario  
e di Emergenza*

Soldà Matteo  
Matricola 1226319

A.A. 2022–2023

# Contents

<b>1</b>	<b>Introduzione</b>	<b>3</b>
1.1	Abstract . . . . .	3
1.2	Problema Generale . . . . .	3
<b>2</b>	<b>Modello</b>	<b>4</b>
2.1	Insiemi . . . . .	4
2.2	Parametri . . . . .	4
2.3	Variabili . . . . .	4
2.4	Funzione Obiettivo . . . . .	4
<b>3</b>	<b>Codice AMPL</b>	<b>6</b>
3.1	File <i>.mod</i> . . . . .	6
<b>4</b>	<b>Scenari</b>	<b>8</b>
4.1	Primo Scenario . . . . .	8
4.1.1	Descrizione . . . . .	8
4.1.2	Codice . . . . .	8
4.1.3	Output . . . . .	9
4.2	Secondo Scenario . . . . .	9
4.2.1	Descrizione . . . . .	9
4.2.2	Codice . . . . .	9
4.2.3	Output . . . . .	10
4.3	Terzo Scenario . . . . .	10
4.3.1	Descrizione . . . . .	10
4.3.2	Codice . . . . .	10
4.3.3	Output . . . . .	10
<b>5</b>	<b>Output su File</b>	<b>11</b>

# 1 Introduzione

## 1.1 Abstract

La Centrale Operativa del 118 della città di Padova necessita di un gran numero di ambulanze per svolgere i trasporti sia ordinari che di emergenza nell'intera provincia, ma con le ambulanze interne e quelle in possesso degli ospedali periferici si riesce a coprire solo in minima parte il fabbisogno: si rende necessario quindi rivolgersi alle Organizzazioni di Volontariato che, in cambio di un rimborso spese, prestano ambulanze e soccorritori per lo svolgimento degli interventi. L'obiettivo del progetto è quello di minimizzare il costo per il mantenimento del servizio, garantendo comunque un pronto intervento in caso di necessità.

## 1.2 Problema Generale

Ogni giorno la Centrale Operativa del 118 di Padova e l'Azienda Ospedaliera ricevono una moltitudine di chiamate, sia per richieste di supporto immediato (che vengono gestite dal SUEM), sia per trasporti programmati (che passano per le linee dell'Azienda Ospedaliera e che non rappresentano casi di emergenza, ma dedicati al trasporto di persone invalide presso l'ospedale per le visite programmate oppure per le dimissioni dal Pronto Soccorso).

L'Azienda Ospedaliera, al suo interno, dispone di un determinato numero di ambulanze di tipo B (dedicate ai trasporti), mentre quelle di tipo A (dedicate al SUEM) sono fornite dagli ospedali periferici. L'insieme di questi mezzi non è però sufficiente a coprire le richieste pervenute dall'intera provincia, ma ne ricopre una minima parte. Per questo motivo, la Centrale Operativa si rivolge a tre Organizzazioni di Volontariato (OdV) che offrono il loro supporto per coprire i viaggi scoperti:

- La Croce Rossa può fornire solo ambulanze di tipo A
- La Croce Bianca può fornire solo ambulanze di tipo B
- La Croce Verde può fornire sia ambulanze di tipo A che di tipo B

Il numero di ambulanze complessive in un determinato giorno viene stimato rispetto alla media di interventi e trasporti effettuati negli stessi giorni delle settimane precedenti, mantenendo comunque un numero arbitrario di ambulanze sia di tipo A che di tipo B in più.

Si vuole quindi minimizzare il costo complessivo settimanale per il servizio, sapendo che:

- Ogni giorno la Centrale Operativa ha necessità di un quantitativo diverso di ambulanze
- Qualora si attivasse una OdV per l'assistenza settimanale, è prevista una singola quota fissa di attivazione
- Ogni ambulanza richiamata dalle OdV prevede un rimborso spese giornaliero che varia in base all'organizzazione e al tipo di ambulanza attivata
- Le ambulanze dell'Azienda Ospedaliera e degli ospedali periferici, se attivate in un determinato giorno, prevedono dei piccoli costi di manutenzione
- Ogni OdV e ospedale dispone di un numero limitato di ambulanze
- Ogni giorno bisogna attivare un numero arbitrario di ambulanze in più rispetto al fabbisogno

## 2 Modello

### 2.1 Insiemi

- *Giorni*: giorni della settimana in cui è attivo il servizio
- *FornitoriA*: fornitori che dispongono di ambulanze di tipo A da attivare
- *FornitoriB*: fornitori che dispongono di ambulanze di tipo B da attivare

### 2.2 Parametri

- *bisognoA<sub>g</sub>*: fabbisogno per il giorno *g* di ambulanze di tipo A (escluso surplus arbitrario)
- *bisognoB<sub>g</sub>*: fabbisogno per il giorno *g* di ambulanze di tipo B (escluso surplus arbitrario)
- *surplusA<sub>g</sub>*: surplus di ambulanze di tipo A per il giorno *g*
- *surplusB<sub>g</sub>*: surplus di ambulanze di tipo B per il giorno *g*
- *maxA<sub>fa</sub>*: numero massimo di ambulanze di tipo A che il fornitore *fa* può fornire in un giorno
- *maxB<sub>fb</sub>*: numero massimo di ambulanze di tipo B che il fornitore *fb* può fornire in un giorno
- *costoGiornalieroA<sub>fa</sub>*: costo per l'attivazione giornaliera per una singola ambulanza di tipo A del fornitore *fa*
- *costoGiornalieroB<sub>fb</sub>*: costo per l'attivazione giornaliera per una singola ambulanza di tipo B del fornitore *fb*
- *costoAttivazioneA<sub>fa</sub>*: costo settimanale per l'attivazione di un fornitore *fa* per la fornitura settimanale di ambulanze di tipo A
- *costoAttivazioneB<sub>fb</sub>*: costo settimanale per l'attivazione di un fornitore *fb* per la fornitura settimanale di ambulanze di tipo B
- *BigM*: utilizzato per vincolare l'utilizzo delle ambulanze rispetto all'attivazione del fornitore (costante sufficientemente grande). Questo, nei file **.dat** è stato definito utilizzando la più vicina potenza del 2 rispetto a  $Max(bisognoA, bisognoB) + Max(surplusA, surplusB)$ .

### 2.3 Variabili

- *ambulanzeA<sub>fa,ga</sub>* = numero di ambulanze del fornitore *fa* attivate il giorno *ga*
- *ambulanzeB<sub>fb,gb</sub>* = numero di ambulanze del fornitore *fb* attivate il giorno *gb*
- $attivazioneSettimanaleA_{fa} = \begin{cases} 1 & \text{se viene attivato il fornitore di ambulanze di tipo A } fa \text{ per la settimana} \\ 0 & \text{altrimenti} \end{cases}$
- $attivazioneSettimanaleB_{fb} = \begin{cases} 1 & \text{se viene attivato il fornitore di ambulanze di tipo B } fb \text{ per la settimana} \\ 0 & \text{altrimenti} \end{cases}$

### 2.4 Funzione Obiettivo

La funzione obiettivo del problema può essere espressa in forma generale come segue:

$$\begin{aligned} \min \quad & \text{Costo Giornaliero Ambulanze Tipo A} + \text{Costo Giornaliero Ambulanze Tipo B} \\ & + \text{Costo Attivazione Ambulanze Tipo A} + \text{Costo Attivazione Ambulanze Tipo B} \end{aligned}$$

Dove i termini sono così definiti:

$$\text{Costo Giornaliero Ambulanze Tipo A} = \sum_{f \in \text{FornitoriA}, g \in \text{Giorni}} \text{ambulanzeA}[f, g] \cdot \text{costoGiornalieroA}[f]$$

$$\text{Costo Giornaliero Ambulanze Tipo B} = \sum_{f \in \text{FornitoriB}, g \in \text{Giorni}} \text{ambulanzeB}[f, g] \cdot \text{costoGiornalieroB}[f]$$

$$\text{Costo Attivazione Ambulanze Tipo A} = \sum_{f \in \text{FornitoriA}} \text{attivazioneSettimanaleA}[f] \cdot \text{costoAttivazioneA}[f]$$

$$\text{Costo Attivazione Ambulanze Tipo B} = \sum_{f \in \text{FornitoriB}} \text{attivazioneSettimanaleB}[f] \cdot \text{costoAttivazioneB}[f]$$

subject to

- Ogni giorno la Centrale Operativa ha necessità di un numero diverso di ambulanze:

$$\sum_{f \in \text{FornitoriA}} \text{ambulanzeA}[f, g] \geq \text{bisognoA}[g] + \text{surplusA}[g] \quad \forall g \in \text{Giorni}$$

$$\sum_{f \in \text{FornitoriB}} \text{ambulanzeB}[f, g] \geq \text{bisognoB}[g] + \text{surplusB}[g] \quad \forall g \in \text{Giorni}$$

- I fornitori dispongono di un numero massimo di ambulanze fornibili in un giorno:

$$\text{ambulanze}[f, g] \leq \text{maxA}[f] \quad \forall f \in \text{FornitoriA}, g \in \text{Giorni}$$

$$\text{ambulanze}[f, g] \leq \text{maxB}[f] \quad \forall f \in \text{FornitoriB}, g \in \text{Giorni}$$

- Le ambulanze di un determinato fornitore si possono attivare se e solo se si è pagata la quota di attivazione settimanale:

$$\text{ambulanzeA}[f, g] \leq \text{BigM} \cdot \text{attivazioneSettimanaleA}[f]$$

$$\text{ambulanzeB}[f, g] \leq \text{BigM} \cdot \text{attivazioneSettimanaleB}[f]$$

**Domini:**

- $\text{ambulanzeA}_{f,g}, \text{ambulanzeB}_{f,g} \geq 0$
- $\text{attivazioneSettimanaleA}_f, \text{attivazioneSettimanaleB}_f \in 0, 1$

## 3 Codice AMPL

### 3.1 File *.mod*

N.B.: il codice ben commentato si trova nella cartella AMPL.

---

```
# =====  
#      Modello AMPL  
#      Ricerca operativa  
#      Solda' Matteo  
#      (2022)  
# =====  
  
### INSIEMI ###  
set Giorni ordered; # Giorni della settimana  
set FornitoriA;    # Fornitori di ambulanze di tipo A  
set FornitoriB;    # Fornitori di ambulanze di tipo B  
  
### PARAMETRI ###  
param bisognoA { Giorni } >= 0 integer default 0; # Fabbisogno giornaliero  
        di ambulanze di tipo A (escluso surplus arbitrario)  
param bisognoB { Giorni } >= 0 integer default 0; # Fabbisogno giornaliero  
        di ambulanze di tipo B (escluso surplus arbitrario)  
param surplusA { Giorni } >= 2 integer default 2; # Ambulanze di tipo A  
        "di scorta"  
param surplusB { Giorni } >= 1 integer default 1; # Ambulanze di tipo B  
        "di scorta"  
param maxA { FornitoriA } >= 0 integer default 15; # Numero massimo di  
        ambulanze di tipo A che un fornitore puo' fornire  
param maxB { FornitoriB } >= 0 integer default 5; # Numero massimo di  
        ambulanze di tipo B che un fornitore puo' fornire  
param costoGiornalieroA { FornitoriA } > 0 default 10; # Costo per  
        l'attivazione giornaliera di una ambulanza di tipo A in base al  
        fornitore  
param costoGiornalieroB { FornitoriB } > 0 default 5; # Costo per  
        l'attivazione giornaliera di una ambulanza di tipo B in base al  
        fornitore  
param costoAttivazioneA { FornitoriA } >= 0 default 100; # Costo per  
        l'attivazione settimanale di una ambulanza di tipo A in base al  
        fornitore (da pagare una sola volta a settimana in caso di attivazione  
        del fornitore)  
param costoAttivazioneB { FornitoriB } >= 0 default 75; # Costo per  
        l'attivazione settimanale di una ambulanza di tipo B in base al  
        fornitore (da pagare una sola volta a settimana in caso di attivazione  
        del fornitore)  
param BigM >= 0 integer default 500;          # BigM per vincoli di tipo  
        logico  
  
### VARIABILI ###  
var ambulanzeA { fa in FornitoriA, ga in Giorni } integer >= 0; # Numero  
        di ambulanze di tipo A fornite dal fornitore fa il giorno ga  
var ambulanzeB { fb in FornitoriB, gb in Giorni } integer >= 0; # Numero  
        di ambulanze di tipo B fornite dal fornitore fb il giorno gb  
var attivazioneSettimanaleA { f in FornitoriA } binary; # Varaibile logica  
        per l'attivazione settimanale delle ambulanze di tipo A di un certo  
        fornitore  
var attivazioneSettimanaleB { f in FornitoriB } binary; # Variabile logica  
        per l'attivazione settimanale delle ambulanze di tipo B di un certo  
        fornitore
```

```

### FUNZIONE OBIETTIVO ###
minimize costo:
    (sum { f in FornitoriA, g in Giorni } ambulanzeA[f, g] *
      costoGiornalieroA[f]) +      # COSTO SETTIMANALE AMBULANZE TIPO A
    (sum { f in FornitoriB, g in Giorni } ambulanzeB[f, g] *
      costoGiornalieroB[f]) +      # COSTO SETTIMANALE AMBULANZE TIPO B
    (sum { f in FornitoriA } attivazioneSettimanaleA[f] *
      costoAttivazioneA[f]) +      # ATTIVAZIONE SETTIMANALE AMBULANZE TIPO A
    (sum { f in FornitoriB } attivazioneSettimanaleB[f] *
      costoAttivazioneB[f])      # ATTIVAZIONE SETTIMANALE AMBULANZE TIPO B
    ;

### VINCOLI ###
# Vincoli Necessita' Giornaliera
subject to necessitaGiornalieraA {g in Giorni} : sum { f in FornitoriA }
    ambulanzeA[f, g] >= bisognoA[g] + surplusA[g];
subject to necessitaGiornalieraB {g in Giorni} : sum { f in FornitoriB }
    ambulanzeB[f, g] >= bisognoB[g] + surplusB[g];

# Vincoli Disponibilita' Fornitori
subject to disponibilitaA { f in FornitoriA, g in Giorni } : ambulanzeA[f,
    g] <= maxA[f];
subject to disponibilitaB { f in FornitoriB, g in Giorni } : ambulanzeB[f,
    g] <= maxB[f];

# Vincoli Logici
subject to attivazioneA { f in FornitoriA, g in Giorni } : ambulanzeA[f,
    g] <= BigM * attivazioneSettimanaleA[f];
subject to attivazioneB { f in FornitoriB, g in Giorni } : ambulanzeB[f,
    g] <= BigM * attivazioneSettimanaleB[f];

```

---

## 4 Scenari

### 4.1 Primo Scenario

#### 4.1.1 Descrizione

In questo primo scenario, i dati sono stati inseriti manualmente, con un numero di ambulanze necessarie tale da permettere di trovare una soluzione ottima.

#### 4.1.2 Codice

---

```
### INSIEMI ###
set Giorni := lunedì martedì mercoledì giovedì venerdì sabato domenica;
set FornitoriA := rossa verde interna;
set FornitoriB := bianca verde interna;

### PARAMETRI ###
param bisognoA :=
lunedì 20 martedì 22 mercoledì 10
giovedì 20 venerdì 30 sabato 30
domenica 15
;

param bisognoB :=
lunedì 3 martedì 5 mercoledì 2
giovedì 3 venerdì 4 sabato 2
domenica 0
;

param surplusA :=
lunedì 3 martedì 3 mercoledì 3
giovedì 2 venerdì 5 sabato 6
domenica 4
;

param surplusB :=
lunedì 1 martedì 3 mercoledì 1
giovedì 1 venerdì 2 sabato 1
domenica 1
;

param maxA :=
rossa 15 verde 20 interna 3
;

param maxB :=
bianca 5 verde 10 interna 8
;

param costoGiornalieroA :=
rossa 15 verde 18 interna 8
;

param costoGiornalieroB :=
bianca 20 verde 12 interna 5
;

param costoAttivazioneA :=
rossa 130 verde 50 interna 0
```



```

;

param costoAttivazioneB :=
bianca 70 verde 55 interna 0
;

param BigM := 64;

```

---

### 4.1.3 Output

L'esecuzione del file **.run** mostra che il costo minimo è di €2929.

## 4.2 Secondo Scenario

### 4.2.1 Descrizione

Il secondo set di dati è stato costruito manualmente con una richiesta di ambulanze non soddisfacibile, impedendo quindi di trovare una soluzione.

### 4.2.2 Codice

---

```

### INSIEMI ###
set Giorni := lunedì martedì mercoledì giovedì venerdì sabato domenica;
set FornitoriA := rossa verde interna;
set FornitoriB := bianca verde interna;

### PARAMETRI ###
param bisognoA :=
lunedì 150 martedì 22 mercoledì 10
giovedì 20 venerdì 30 sabato 30
domenica 15
;

param bisognoB :=
lunedì 65 martedì 5 mercoledì 2
giovedì 3 venerdì 4 sabato 2
domenica 0
;

param surplusA :=
lunedì 3 martedì 3 mercoledì 3
giovedì 2 venerdì 5 sabato 6
domenica 4
;

param surplusB :=
lunedì 1 martedì 3 mercoledì 1
giovedì 1 venerdì 2 sabato 1
domenica 1
;

param maxA :=
rossa 15 verde 20 interna 3
;

param maxB :=
bianca 5 verde 10 interna 8
;

```

```

param costoGiornalieroA :=
rossa 15 verde 18 interna 8
;

param costoGiornalieroB :=
bianca 20 verde 12 interna 5
;

param costoAttivazioneA :=
rossa 130 verde 50 interna 0
;

param costoAttivazioneB :=
bianca 70 verde 55 interna 0
;

param BigM := 256;

```

---

### 4.2.3 Output

Ovviamente, il costo minimo trovato eseguendo il file **.run** sarà pari a €0.

## 4.3 Terzo Scenario

### 4.3.1 Descrizione

Nel terzo scenario sono stati definiti soltanto gli insiemi, così da poter sfruttare i valori di default.

### 4.3.2 Codice

---

```

### INSIEMI ###
set Giorni := lunedì martedì mercoledì giovedì venerdì sabato domenica;
set FornitoriA := rossa verde interna;
set FornitoriB := bianca verde interna;

### PARAMETRI ###
# Non definiti per test valori di default

```

---

### 4.3.3 Output

Eseguendo il file **.run** dedicato, utilizzando quindi i dati di default, il costo minimo risulta essere di €350.

## 5 Output su File

All'interno della cartella **AMPL** si può trovare un file denominato *Ospedale.FileSave.run* che esegue in serie tutti e tre gli scenari sopracitati e ne salva l'output in un file dedicato chiamato *Output.txt*. Per fare ciò, sono stati utilizzati gli operatori `>` e `>>`. Il primo serve a creare un nuovo file (o a ripulirlo se già esistente) e scriverci l'istruzione della stessa linea; il secondo serve invece per aprire il file e fare un *append* dell'istruzione di riferimento.